

GB007: 2次元誘電体

本 whitepaper は Gunnar Backstrom 氏の承諾のもと、書籍“*Simple Fields of Physics by Finite Element Analysis*” に記されている多数の FlexPDE 適用事例の中からその一部を紹介するものです。

誘電体 (dielectrics; 電場を加えたときに分極を生じる物質) を対象とする場合には電束密度 (electric displacement) ベクトル

$$\mathbf{D} = \varepsilon \mathbf{E} = \varepsilon_r \varepsilon_0 \mathbf{E}$$

を含む関係式を扱う必要があります。ここに ε_r は比誘電率 (relative permittivity) を意味します。この場合のマックスウェルの方程式は

$$\nabla \cdot \mathbf{D} = \rho_v \quad (1)$$

で与えられることとなります。ここに ρ_v は真電荷 (free charge; 自由電荷) 密度 (分極から生じたものではない電荷密度) を表すパラメータです。ここで扱う問題においては ρ_v が 0 のケースのみを考えるので、FlexPDE に与える方程式は $\text{div}(\mathbf{D}) = 0$ となります。

数式 (1) はポテンシャル U に関するポアソン方程式


$$\nabla \cdot \mathbf{D} = \nabla \cdot (\varepsilon \mathbf{E}) = \nabla \cdot (-\varepsilon \nabla U) = \rho_v \quad (2)$$

に展開されることとなります。

1 Case1 - 同軸ケーブル

右の図は同軸ケーブルの断面を示したものです（単位としては SI 単位系を使用）。外側のシールドの電位を 0、中心線の電位を 1 とし、その中間に位置するポリマー部分における場を調べることにします。なお、中心線の半径は $2e-3\text{ m}$ 、ポリマーの比誘電率は 2.3 とします。

なお、電束密度ベクトルを D としたとき、その x 成分、 y 成分を D_x, D_y と表現したくなりますが、FlexPDE の場合、これらは微分演算子を表す予約語 (dx, dy) です。このため以下のスクリプト中では D_x, D_y という表現が使用されている点に注意してください。



1.1 Problem descriptor [dielectrics01a.pde]

まずタイトルを設定します。

```
TITLE
'Coaxial Cable'      { dielectrics01a.pde }
```

次に演算精度に関するセレクトをセットします。

```
SELECT
Errlim = 3e-4
```

従属変数を定義します。

```
VARIABLES
U      { Electric potential }
```

偏微分方程式の定義に先立ち、パラメータ類を SI 単位系で定義します。なお、 cap_ex という数式は円筒型コンデンサの容量を与える理論式であり、Elevation プロット中で使用されます。

```
DEFINITIONS      { SI units }
r1 = 2e-3  r2 = 1e-2  U1 = 1.0
eps0 = 8.854e-12    { Permittivity of vacuum }
eps = 2.3*eps0      { Permittivity of polymer }
Ex = -dx(U)  Ey = -dy(U)  E = -grad(U)  Em = magnitude(E)
D_x = eps*Ex  D_y = eps*Ey  D = eps*E  Dm = magnitude(D)
                                     { Dielectric displacement }
cap_ex = 2*PI*eps/ln(r2/r1) { Theoretical capacitance }
```

次に方程式を定義します。真電荷が存在しないケースを想定しているので右辺は 0 となります。

```
EQUATIONS
  div(D) = 0           { No volume charge }
```

境界の形状と境界条件（共に Dirichlet 型）を設定します。内部境界内は解析対象ドメインから除外されます。

```
BOUNDARIES
  Region 1
    Start 'outer' (r2, 0)
      Value(U) = 0   Arc(Center = 0,0) Angle = 360
    Start 'inner' (r1, 0)
      Value(U) = U1  Arc(Center = 0,0) Angle = 360 { Cut-out }
```

最後に出力すべき情報を規定します。


```
PLOTS
  Grid(x, y)
  Contour(U)
  Vector(D) norm
  Contour(D_x)  Contour(D_y)
  Contour(Dm)  Surface(Dm)
  Elevation(Dm) on 'inner'
    Report(cap_ex) as 'Theoretical capacitance'
  Elevation(Dm) on 'outer'
```

```
END
```

1.2 実行結果


(1) Grid(x, y)

FlexPDEによって生成されたメッシュ構成を示しています。メッシュ再構成は1回行われ、内周部のメッシュが細分化されています。




(2) Contour(U)

等電位線のプロットです。内周境界で $U = 1$ 、外周境界で $U = 0$ となっています。




(3) Vector(D) norm

電束密度 D のベクトル場をプロットしたものです。中心から放射状に広がるベクトル場が描かれています。
norm を指定しているため、ベクトル場の大きさは色で判断することになります。




(4) Contour(D_x)

電束密度ベクトル D の x 成分 D_x の値を等高線の形でプロットしたものです。




(5) Contour(D_y)

電束密度ベクトル D の y 成分 D_y の値を等高線の形でプロットしたものです。


(6) Contour(D_m)

電束密度ベクトル D の絶対値を等高線の形でプロットしたものです。ポテンシャルの勾配が大きい内周部で大きな値となっています。




(7) Surface(Dm)

電束密度ベクトル D の絶対値を曲面図の形でプロットしたものです。



(8) Elevation(Dm) on 'inner' Report(cap_ex)

内周境界に沿って電束密度ベクトル D の絶対値がどう変化しているかをグラフ化したものです。概ね $6.3e-9$ 前後の値となっており、それを境界上で積分した値が $7.92e-11$ と算出されています。



円筒形のコンデンサの容量 C は

$$C = \frac{2\pi\epsilon}{\ln(r_2/r_1)} \quad (3)$$

で与えられます (スクリプト上の cap_ex)。この値は Report 文の形で出力されており、 $7.95e-11$ という値となっています。今、内側と外側の電極間の電位差は 1V であるので、同軸ケーブル単位長当りの真電荷量の理論値は $Q_F = CU = C = 7.95e-11$ と算出されます。

ここで、内周境界である半径 r_1 の円筒状曲面 (長さは単位長 1) S についてガウスの定理を適用すると、上記数式 (1) に注意して

$$\iint_S \mathbf{D} \cdot \mathbf{n} dS = \iiint_V \nabla \cdot \mathbf{D} dV = \iiint_V \rho_v dV = Q_F$$


が導かれます。一方、ベクトル \mathbf{D} は円周に対して垂直ですから

$$\iint_S \mathbf{D} \cdot \mathbf{n} dS = \iint_S |\mathbf{D}| dS$$

として計算できるわけですが、この値は $7.92e-11$ としてプロット下部に出力されています。FlexPDE による $|\mathbf{D}|$ の計算値にはかなりの変動が見られるわけですが、それでも理論値 $7.95e-11$ に対しては誤差 0.4% 以内に収まっています。

(9) Elevation(Dm) on 'outer'


外周境界に沿って電束密度ベクトル \mathbf{D} の絶対値がどう変化しているかをグラフ化したものです。概ね $1.263e-9$ 前後の値となっており、それを境界上で積分した値は $7.94e-11$ と算出されています。



ここで示したスクリプト上では誘電体の誘電率 (eps) として一定値を用いましたが、例えば中心からの半径によって誘電率が変化するケース等についてもわずかな変更で対処できます。

2 Case2 - 平行板コンデンサ

今度は平行板コンデンサ内、及び周囲の場について考察します。右の図は2枚の金属板の間に誘電体が入り込んだ形の平行板コンデンサの断面図です。上側のプレートには $1.0e4V$ の電位をかけ、下側のプレートは接地するものとします。 $2m \times 2m$ のドメイン周囲は絶縁境界であるとして計算を実行します。なおプレートの横幅は $0.5m$ 、プレート間の幅は $0.2m$ 、誘電体の比誘電率は 7.0 と設定します。



2.1 Problem descriptor [dielectrics01b.pde]

基本形は Case1 の dielectrics01a.pde と変わりません。

```
TITLE
  'Parallel Plate Capacitor'    { dielectrics01b.pde }

SELECT
  Errlim = 3e-4

VARIABLES
  U                               { Electric potential }
```

偏微分方程式の定義に先立ち、パラメータ類を SI 単位系で定義します。この段階ではデフォルト値の意味で $\varepsilon = \varepsilon_0$ と設定していますが、誘電体のリージョン内においては $\varepsilon = 7.0\varepsilon_0$ という形でパラメータ値の変更を行います。なお、結果の検証用に平行板コンデンサの容量を計算する数式 cap_ex を定義しておきます。この数式は Elevation プロット中で使用されます。

```

DEFINITIONS          { SI units }
L = 1  xx = 0.25  d0 = 0.1  dd0 = 0.4*d0
U0 = 1.0e4          { Voltage }
eps0 = 8.854e-12   { Permittivity of vacuum }
eps = eps0         { Permittivity, default }
Ex = -dx(U)  Ey = -dy(U)  E = -grad(U)  Em = magnitude(E)
D_x = eps*Ex  D_y = eps*Ey  D = eps*E  Dm = magnitude(D)
          { Dielectric displacement }
cap_ex = 7.0*eps0*(2*xx)/(2*d0) { Theoretical capacitance }

```

次に方程式を定義します。解析対象のドメインからは上下の金属板は exclude されている点に注意してください。このドメイン内において真電荷は存在しないので右辺は 0 となります。

```

EQUATIONS
div(D) = 0          { No total charge }

```

境界の形状と境界条件を設定します。上下の金属板は解析対象ドメインから除外されます。一方、誘電体部分は別個のリージョンとして定義し、固有の誘電率を設定します。

```

BOUNDARIES
Region 1
Start 'outer' (-L, -L) Natural(U) = 0 { Outer boundary }
Line to (L, -L) to (L, L) to (-L, L) to Close
Start 'upper' (-xx, d0) Value(U) = U0 { Upper plate }
Line to (xx, d0) to (xx, d0 + dd0) to (-xx, d0 + dd0) to Close
Start 'lower' (-xx, -d0 - dd0) Value(U) = 0.0
          { Lower plate, ground }
Line to (xx, -d0 - dd0) to (xx, -d0) to (-xx, -d0) to Close

Region 'glass' eps = 7.0*eps0 { Dielectric }
Start (-xx, -d0) Line to (xx, -d0) to (xx, d0)
          to (-xx, d0) to Close

```

最後に出力すべき情報を規定します。

```

PLOTS
  Grid(x, y)
  Contour(U)
  Vector(D) norm
  Contour(Dm) Surface(Dm)
  Contour(Dm) log
  Contour(Dm) painted on 'glass'
  Elevation(D_y) from (-L, 0) to (L, 0)
  Report(cap_ex) as 'Theoretical capacitance'


END

```

2.2 実行結果


(1) Grid(x, y)

指定された演算精度を達成するためにメッシュの細分化が2度ほど実施されたことが“Grid#3”という表示からわかります。




(2) Contour(U)

等電位線のプロットです。金属板にはさまれた誘電体内では等電位線がほぼ平行に走っていることが確認できます。なお、周囲は絶縁境界なので等電位線は境界と直交する形となっています。




(3) Vector(D) norm

電束密度 D のベクトルプロットです。上の金属板から出た電気力線はすべて下の金属板に流れ込んでいる様子が示されています。




(4) Contour(Dm)

電束密度ベクトル D の絶対値を等高線の形でプロットしたものです。ドメイン内の大半の領域で $|D|$ の値が一定であるために、等高線では表現しにくい関数形となっています。




(5) Surface(Dm)

電束密度ベクトル D の絶対値を曲面図の形でプロットしたものです。誘電体内、及びその周囲で平面に近い形状となっているため等高線での表現には不向きということがわかります。




(6) Contour(Dm) log

電束密度ベクトル D の絶対値に関する等高線図ですが、対数スケールでプロットしたものです。それでも誘電体内には等高線がほとんど現れてきません。




(7) Contour(Dm) painted on 'glass'

誘電体内にしばって $|D|$ の等高線図を描いたものです。外部と接する部分を除くとほぼ平坦な形状となっていることがわかります。




(8) Elevation(D_y) from $(-L, 0)$ to $(L, 0)$ Report(cap_ex)

電束密度ベクトル D の y 成分 D_y の値が x 軸上でどう変化しているかをプロットしたものです。積分値である $-1.68e-6$ という値については少々説明を加える必要があるでしょう。



下の金属板を含む周回路 $\Gamma_1, \Gamma_2, \Gamma_3, \Gamma_4$ に沿って D の法線成分を積分することを考えます (下図参照)。この場合、 $\Gamma_2, \Gamma_3, \Gamma_4$ 上では法線成分が 0 となるため、積分に寄与するのは Γ_1 上での積分のみとなります。



この周回路を z 軸方向に単位長のばした直方体の表面を S としたとき、ガウスの定理と数式 (1) より

$$\iint_S \mathbf{D} \cdot \mathbf{n} dS = \iiint_V \nabla \cdot \mathbf{D} dV = \iiint_V \rho_v dV = Q_F$$

が言えます (Q_F は金属板単位長当りの真電荷量)。一方、

$$\iint_S \mathbf{D} \cdot \mathbf{n} dS = \iint_S |\mathbf{D}| dS = \iint_{\Gamma_1} |\mathbf{D}| dS$$

より、FlexPDE が出力してきた $-1.68e-6$ という値が Q_F を意味することになります。 $Q = CU$ という等式にこの Q_F の値と $U = 1.0e4$ を代入するとコンデンサの容量が $1.68e-10$ と算出されます。

一方、Report 文で平行板コンデンサ容量の理論値を出力させていますが、その値は $1.55e-10$ というところで FlexPDE の計算値とは若干異なります。これは後者が誘電体左右端での場の乱れを想定していないことによるものです。

3 Case3 - 平行場中のガラス棒


今度は右図に示されるように、平行な電場の中に断面が円形のガラス棒を置いたときの場の変化について考察します。この場合にポイントとなるのはドメインの左側境界と右側境界において

$$\frac{\partial U}{\partial n}$$

すなわちポテンシャルの法線微分値を Natural 文によって指定することになるわけですが、方程式が

$$\operatorname{div}(\mathbf{D}) = 0$$

という形式で与えられているため、 U には ε がかかっていると考える必要があります。このため境界条件で法線微分値を指定する際にも ε がかった形で指定することになるのでご注意ください。これは部分積分からもたらされる要請と言えます。



3.1 Problem descriptor [dielectrics01c.pde]

Case2 の dielectrics01b.pde をベースにスクリプトを用意します。

```
TITLE
  'Glass Rod across a Parallel Field'    { dielectrics01c.pde }

SELECT
  Errlim = 3e-5
```

VARIABLES

```
U { Electric potential }
```

偏微分方程式の定義に先立ち、パラメータ類を SI 単位系で定義します。この段階では ϵ の値は設定されていませんが、リージョン定義の中で具体的な値を指定しています。

DEFINITIONS

```
{ SI units }
L = 1 r0 = 0.2
eps0 = 8.854e-12 { Permittivity of vacuum }
epsr2 = 7.0 { Relative permittivity of glass }
eps { Declared only }
Ex0 = 1e4 { Far field }
Ex = -dx(U) Ey = -dy(U) E = -grad(U) Em = magnitude(E)
D_x = eps*Ex D_y = eps*Ey D = eps*E Dm = magnitude(D)
{ Dielectric displacement }
```

次に方程式を定義します。解析対象のドメイン内には真電荷は存在しないので右辺は 0 となります。

EQUATIONS

```
div(D) = 0 { div(eps*(-grad(U))) = 0 }
```

境界の形状と境界条件を設定します。法線微分値のみの指定では従属変数値が一意に決まらないため (GB001 参照) 下辺上で Point value の指定を入れている点に注意してください。また左辺の法線微分値はドメインから見て内向きとなるため負号を付ける必要があります。ここでは 'domain' と 'glass' という 2 つのリージョンを設定、その各々で別個の誘電率を規定していますが、これらリージョン名は境界上での積分計算を行う際に重要な意味を持ってきます。

BOUNDARIES

```
Region 'domain' eps = eps0 { Vacuum outside }
Start 'outer' (-L, -L)
Natural(U) = 0 Line to (0, -L) Point value(U) = 0
Line to (L, -L)
Natural(U) = eps*Ex0 Line to (L, L)
Natural(U) = 0 Line to (-L, L)
Natural(U) = -eps*Ex0 Line to Close

Region 'glass' eps = epsr2*eps0 { Glass }
Start 'circle' (r0, 0) Arc(Center = 0,0) Angle = 360
```

最後に出力すべき情報を規定します。

```

PLOTS
  Grid(x, y)
  Contour(U)
  Elevation(U) on 'outer'
  Elevation(normal(D)) on 'outer' Report(eps0*Ex0)
  Vector(D) norm   Vector(D) on 'glass'
  Elevation(tangential(E)) on 'circle' on Region 'domain'
  Elevation(tangential(E)) on 'circle' on Region 'glass'
  Elevation(normal(D)) on 'circle' on Region 'domain'
  Elevation(normal(D)) on 'circle' on Region 'glass'
  Elevation(normal(E)) on 'circle' on Region 'domain'
  Elevation(normal(E)) on 'circle' on Region 'glass'


END

```

3.2 実行結果


(1) Grid(x, y)

FlexPDE によって自動生成された有限要素メッシュの構成を示しています。




(2) Contour(U)

等電位線を表す U に関する等高線図です。




(3) Elevation(U) on 'outer'

外周境界上で U の値がどう変化しているかを示しています。左右の境界上における電位はガラス棒の存在によりわずかながら変化している点にご注意ください。




(4) Elevation(normal(D)) on 'outer' Report(eps0*Ex0)

電束密度ベクトル D の外周上での法線成分の値をプロットしたものです。左右の境界上で取るべき値は Report 文によりプロット下部に出力されています。Natural 文で指定した境界条件が正しく反映されていることが確認できます。




(5) Vector(D) norm

電束密度 D のベクトルプロットです。左から右への流れが表現されています。




(6) Vector(D) on 'glass'

誘電体内での D に関するベクトルプロットです。全体として平行なベクトル場となっており、またベクトルの大きさもほぼ一様であることが右の表からわかります。




(7) Elevation(tangential(E)) on 'circle' on Region 'domain'

内部境界 'circle' 上で電場ベクトル E の接線方向成分の値をプロットしたものです。ただし境界線はリージョン 'domain' とリージョン 'glass' 双方に属するため、そのどちらに視点を置かによって符号が変わってきます。(8) は 'glass' 側でプロットしているためグラフは反転しています。





右の図は境界線をはさんで隣接するセルを示しています。矢印はセル境界をトレースする際に用いられる反時計回りの向きを示すものですが、境界線上では対向したものとなるため、接線方向成分の値の取り方についてもこれらセル間では符号が逆になります。




FlexPDE 5.0.22 の段階では 'domain' 側から見たときの tangential の符号が逆になるというバグがあります。このため (7) のプロットは符号を反転した形で出力しています。次のリリースでは修正される予定です。

(8) Elevation(tangential(E)) on 'circle' on Region 'glass'

内部境界 'circle' 上で電場ベクトル E の接線方向成分の値をリージョン 'glass' 側からプロットしたものです。(7) の符号を反転させれば (8) と重なったものとなります。これは


電場の接線成分は境界面において連続である

という理論上の要請に基づくものである点に注意してください。




(9) Elevation(normal(D)) on 'circle' on Region 'domain'

内部境界'circle'上で電束密度ベクトル D の法線方向成分の値をリージョン'domain'側からプロットしたものです。



右の図は境界線をはさんで隣接するセルを示しています。境界線上での法線成分はリージョンから外に向かう方向が正として扱われます。




FlexPDE 5.0.22 の段階では'domain'側から見たときの normal の符号が逆になるというバグがあります。このため (9) のプロットは符号を反転した形で出力しています。次のリリースでは修正される予定です。

(10) Elevation(normal(D)) on 'circle' on Region 'glass'

内部境界'circle'上で電束密度ベクトル D の法線方向成分の値をリージョン'glass'側からプロットしたものです。(9)の符号を反転させれば(10)と重なったものとなります。これは


電束密度の法線成分は境界面において連続である

という理論上の要請に基づくものである点に注意してください。




(11) Elevation(normal(E)) on 'circle' on Region 'domain'

同じ法線成分であっても電場ベクトル E の場合には境界の両側で連続とはなりません。(12) とはスケールが異なる点に注意してください。



(12) Elevation(normal(E)) on 'circle' on Region 'glass'

電場ベクトル E の法線成分の値をリージョン 'glass' 側からプロットしたものです。符号を反転させてもスケールが異なるため、(11) のグラフとは重なりません。



4 Case4 - 誘電分極

Case3 と全く同じ問題ですが、ここではガラス棒表面に生ずる分極電荷密度(polarization charge density) を計算してみましょう。単位体積当りの電気双極子モーメント (dipole moment) を P とすると

$$P = D - \varepsilon_0 E = \left(1 - \frac{\varepsilon_0}{\varepsilon}\right) D$$


で定義されます。このとき、分極電荷は

$$\rho_p = -\nabla \cdot P = -\nabla \cdot \left(1 - \frac{\varepsilon_0}{\varepsilon}\right) D$$

で、表面分極電荷密度は

$$\sigma_{ps} = P \cdot n = \left(1 - \frac{\varepsilon_0}{\varepsilon}\right) D_n$$

のように計算することができます。



4.1 Problem descriptor [dielectrics01d.pde]

Case3 で作成したスクリプト dielectrics01c.pde に対する修正分のみを記すことにします。

TITLE

```
'Glass Rod across a Parallel Field, Polarization'
{ dielectrics01d.pde }
```

DEFINITIONS セクションにおいて次の 1 文を追加します。

```
DEFINITIONS { SI units }
...
P = (1 - eps0/eps)*D { Dipole moment }
```

半円周上での積分値を計算するため、Feature 文を使って開曲線を定義します。併せて 'circle' の定義も若干変更します。

BOUNDARIES

```
...
Region 'glass' eps = epsr2*eps0 { Glass }
Start 'circle' (-r0, 0) Arc(Center = 0,0) Angle = 360

Feature 'righthalf'
Start (0, -r0) Arc(Center = 0,0) Angle = 180
```

ここでは分極に関する情報のみをプロットすることにします。

PLOTS


```
Elevation(normal(P)) on 'circle' on Region 'glass'
      as 'Polarization surface charge'
Elevation(normal(P)) on 'righthalf' on Region 'glass'
Elevation(normal(P)) on 'circle' on Region 'domain'
```

END

4.2 実行結果


(1) Elevation(normal(P)) on 'circle' on Region 'glass'

ガラス棒の円周に沿う形で電気分極 P の法線成分をプロットしたものです。なお、円周全体にわたって積分したのでは積分値は 0 となってしまうので、円周右半分 (パス名 'righthalf') についての Elevation プロットを別途作成したのが次の (2) です。



(2) Elevation(normal(P)) on 'righthalf' on Region 'glass'

プロット下部に示されている $5.18e-8$ という値は電気分極 P の法線成分を右側の半円周上で積分した値を示しています。



(3) Elevation(normal(P)) on 'circle' on Region 'domain'

同じ内容の elevation プロットですが、リージョン 'domain' 側では値が 0 となっています。

